1. ПОСТАНОВКА ЗАДАЧИ

Написать программу для автоматического построения детерминированного конечного автомата (ДКА), эквивалентного заданной регулярной грамматике. 

Вход программы: терминальный и нетерминальный алфавиты грамматики, целевой символ, правила грамматики, цепочки для распознавания. 

Выход: построенный ДКА (все 5 элементов), результат проверки цепочек.

Подробно:

Язык задан регулярной грамматикой, причём она может быть не автоматного вида. При написании программы разработчику разрешается выбрать один из двух типов регулярной грамматики (ЛЛ или ПЛ) и следует информировать об этом пользователя. Терминальный алфавит грамматики может включать в себя любые символы, в нетерминальном алфавите могут использоваться заглавные буквы латинского алфавита или (на усмотрение разработчика) слова. Правила задаваемой грамматики должны соответствовать выбранному типу. Для того чтобы в исходной грамматике можно было использовать пустое правило, необходимо предусмотреть поле ввода для символа, которым пользователь может обозначить пустую цепочку.

Программа должна:

1. по заданной регулярной грамматике строить эквивалентный ДКА, распознающий этот же язык, в том виде, как он рассматривался в теории, раздел 2.2.2;

2. с помощью построенного ДКА проверять вводимые пользователем цепочки на их принадлежность этому языку.

ДКА должен распознавать язык, задаваемый исходной грамматикой, т.е. являться эквивалентной конструкцией. Функция переходов ДКА может изображаться в виде таблицы или графа, вариант вида её представления выбирается разработчиком. Для удобства построения автомата рекомендуется предварительно привести заданную грамматику к автоматному виду (в соответствии с лекционным разделом 2.2.1).

При выборе такого способа построения ДКА, когда сначала по заданной грамматике строится эквивалентный НКА, а затем он приводится к детерминированному виду, промежуточный результат в виде НКА необходимо также отображать на экране по просьбе пользователя.

После построения ДКА пользователь может вводить произвольные цепочки для проверки их на принадлежность исходному языку. Разбор цепочек автоматом следует поэтапно отображать на экране в виде последовательной смены конфигураций в соответствии с лабораторной работой №2.

2. ОПИСАНИЕ АЛГОРИТМА РЕШЕНИЯ ЗАДАЧИ

При выполнении курсового проекта в качестве типа исходной регулярной грамматики была выбрана леволинейная грамматика. В соответствии с постановкой задачи нетерминальными символами грамматики служат заглавные символы латинского алфавита, что ограничивает их число 26 символами. Соответственно, в нетерминальный алфавит могут входить любые другие символы, не совпадающие с заглавными символами латинского алфавита и с символом, выбранным в качестве пустой цепочки.

Построение ДКА по исходной грамматике начинается с проверки корректности ее задания, в том числе проверяется соответствие всех ее правил допустимой структуре правил леволинейной грамматики. Далее, в соответствии с лекционным разделом 2.2.1 исходная ЛЛ-грамматика приводится к автоматному виду. 

По полученной автоматной грамматике выполняется построение НКА:

· множество нетерминальных символов грамматики становится множеством состояний автомата;

· во множество состояний автомата добавляется новое состояние, которое объявляется начальным состоянием автомата;

· начальный символ грамматики заносится во множество конечных состояний НКА;

· для всех правил вида N1(N2t в автомат добавляется переход из N2 в N1 по символу ‘t’;

· для всех правил вида N(t в автомат добавляется переход из начального состояния НКА в N по символу ‘t’;

· для всех правил вида N(( состояние N объявляется начальным. Для того, чтобы избежать двух начальных состояний в НКА, данное состояние объединяется с предыдущим начальным состоянием и добавляется в автомат как новое начальное состояние автомата. Если N совпадает с целевым символом автоматной грамматики, то вместо объединения двух состояний НКА, выполняется только объявление начального состояния автомата его конечным состоянием.

Построенный НКА проверяется на детерминированность путем проверки отсутствия в нем состояний, из которого возможен переход по одинаковому терминальному символу в несколько разных состояний. Если таких состояний   нет, то построенный автомат является детерминированным и его дальнейшие преобразования не требуются.

Если построенный НКА не является детерминированным, то он преобразуется в эквивалентный ДКА в соответствии с алгоритмом, изложенном в лекционном разделе 2.2.2.

Минимизируем построенный ДКА, удаляя из него недостижимые состояния.

Вы программу даже не запускали? Или Вас это вот никак не смутило:

Вроде бы, на титульном листе другая фамилия… 
Поскольку программа чужая, то для получения положительной оценки придётся выполнить дополнительное задание. Добавьте возможность задавать исходную грамматику в праволинейном виде.  
Пример.

Построение эквивалентного ДКА по ЛЛ-грамматике

G({a,b,c},{S,A},P,S)

правила Р:
S(Acc

A(Aa

A(Ab

A(ac

A(ab

Приведение ЛЛ-грамматики к автоматному виду

· выполняются следующие назначения для автоматной грамматики: VN={S,A}, VT={a,b,c}, целевой символ равен S;

· правило S(Acc преобразуется в два правила S(Bc и B(Ac, в автоматную грамматику добавляется нетерминальный символ B: VN={S,A,B};

· правила A(Aa и A(Ab  соответствуют правилам автоматной грамматики и переносится в ее правила без изменения;

· правило A(ac преобразуется в два правила A(Cc и C(a, в автоматную грамматику добавляется нетерминальный символ C: VN={S,A,B,C};

· правило A(ab преобразуется в два правила A(Db и D(a, в автоматную грамматику добавляется нетерминальный символ D: VN={S,A,B,C,D}.

Построена автоматная грамматика, эквивалентная исходной грамматике:

G({a,b,c}, {S,A,B,C,D},P,S)

правила Р:

S(Bc
B(Ac
A(Aa

A(Ab

A(Cc

C(a

A(Db
D(a

Построение НКА, эквивалентного автоматной грамматике

· множество состояний автомата {S,A,B,C,D};

· добавляется начальное состояние автомата E;

· состояние S заносится во множество конечных состояний автомата;

· правило S(Bс добавляет переход из B в S по символу ‘c’;

· правило B(Aс добавляет переход из A в B по символу ‘c’;

· правило A(Aa добавляет переход из A в A по символу ‘a’;

· правило A(Ab добавляет переход из A в A по символу ‘b’;

· правило A(Cc добавляет переход из C в A по символу ‘c’;

· правило C(a  добавляет переход из E в C по символу ‘a’;

· правило A(Db добавляет переход из D в A по символу ‘b’;

· правило D(a  добавляет переход из E в D по символу ‘a’.

Построен эквивалентный автоматной грамматике НКА:

M({S,A,B,C,D,E},{a,b,c},δ,E,{S})

функция переходов δ:

Данный автомата не является детерминированным, т.к. имеет переходы из состояния E в C и D по одному терминальному символу ‘a’.


Построение эквивалентного ДКА

	
	a
	b

В таблице есть переходы из состояния E в C и D по одному терминальному символу ‘a’. Добавляем в автомат новое состояние CD и соответствующие переходы:

	
	a

	A
	{A}

В таблице переходов больше нет состояний, из которых есть переходы в разные состояние по одному терминальному символу. Автомат является детерминированным.


Удаление недостижимых состояний

	
	a
	b
	c

	S
	
	
	

	A

	{A}

	E
	{CD}

В данном ДКА состояния C и D недостижимы из начального состояния E. Удаляем их из автомата, одновременно заменяя обозначение состояния CD на освободившееся состояние C.

Построен ДКА, эквивалентный исходной ЛЛ-грамматике:

M({S,A,B,C,E},{a,b,c},δ,E,{S})

функция переходов δ:

	
	a
	b
	c

	S
	
	
	

	A
	{A}
	{A}
	{B}

	B
	
	
	{S}

	E
	{C}
	
	

	C
	
	{A}
	{A}


Входные данные вводятся в программу с клавиатуры или из текстового файла, пример заполнения которого содержится в файле ‘LL.TXT’. При обнаружении ошибок в входном файле программа автоматически выводит на экран подсказку с описанием формата задания ЛЛ-грамматики. Эта же подсказка может быть вызвана через меню ‘Помощь’. 

Чтение цепочки построенным ДКА отображается на экране в поле озаглавленном ‘Такты автомата’ в виде смены конфигураций в соответствии с лабораторной работой №2.

3. ОПИСАНИЕ ОСНОВНЫХ ПЕРЕМЕННЫХ И БЛОКОВ

Глобальные переменные программы:

AVN – нетерминальные символы автоматной грамматики;

Edit1 – нетерминальные символы исходной грамматики;

Edit2 – терминальные символы исходной грамматики;

Edit3 – целевой символ исходной грамматики;

NVN – число состояний автомата;

S0  – начальное состояние автомата;

StringGrid1 – таблица переходов автомата;

VN –
состояния автомата.

Основные программные блоки:

– Auto –  преобразование ЛЛ-грамматики в автоматную грамматику;

– CheckDKA –  проверка автомата на детерминированность;

– DeleteUnavailables – удаление недостижимых состояний автомата;

– DKA –  преобразование НКА в ДКА;

– NKA –  построение НКА по автоматной грамматике.

4. ТЕКСТ ПРОГРАММЫ
Unit1.pas

unit Unit1;

{$R+}

interface

uses

  Windows, Messages, SysUtils, Variants, Classes, Graphics, Controls, Forms,

  Dialogs, Math, StdCtrls, Grids, Menus, ExtCtrls;

type

  TForm1 = class(TForm)

    Label1: TLabel;

    Label2: TLabel;

    Label3: TLabel;

    Label4: TLabel;

    Label5: TLabel;

    Edit1: TEdit;

    Edit2: TEdit;

    Edit3: TEdit;

    StringGrid1: TStringGrid;

    MainMenu1: TMainMenu;

    N1: TMenuItem;

    N2: TMenuItem;

    N3: TMenuItem;

    Edit4: TEdit;

    Edit6: TEdit;

    Edit7: TEdit;

    Label6: TLabel;

    N4: TMenuItem;

    N5: TMenuItem;

    Label7: TLabel;

    Label10: TLabel;

    Edit9: TEdit;

    N6: TMenuItem;

    N7: TMenuItem;

    OpenDialog1: TOpenDialog;

    SaveDialog1: TSaveDialog;

    Memo1: TMemo;

    ListBox1: TListBox;

    N9: TMenuItem;

    Label8: TLabel;

    ListBox2: TListBox;

    Button1: TButton;

    Edit5: TEdit;

    procedure N3Click(Sender: TObject);

    procedure N4Click(Sender: TObject);

    procedure N2Click(Sender: TObject);

    procedure N5Click(Sender: TObject);

    procedure Button1Click(Sender: TObject);

    procedure N6Click(Sender: TObject);

    procedure N7Click(Sender: TObject);

    procedure N9Click(Sender: TObject);

  private

    function AddAVN(var V: String): Boolean;

    function AddVN(P, S: String): String;

    function CheckDKA: Boolean;

    function CheckState(S: String): Boolean;

    procedure DeleteUnavailables;

    procedure DKA;

    procedure Out;

    function Auto: Boolean;

    function NKA: Boolean;

    function SortString(S: String): String;

  public

    { Public declarations }

  end;

  TVN = Array [1..26] of String;

var

 Form1: TForm1;

 AVN: String;

 NVN: Integer;

 S0: String;

 VN: TVN;

 P: Array [1..26,1..10] of String;

 Final: Array [1..26] of Boolean;

implementation

{$R *.dfm}

function TForm1.AddVN(P, S: String): String;

var I: Integer;

begin

 for I:=1 to Length(S) do

  if Pos(S[I],P)=0 then P:=P+S[I];

 AddVN:=SortString(P);

end;

function TForm1.CheckDKA: Boolean;

var I, J: Integer;

begin

 for I:=1 to NVN do

  for J:=1 to Length(Edit2.Text) do

   if Length(P[I,J])>1 then begin

    Result:=false;

    exit;

   end;

 Result:=true;

end;

function TForm1.CheckState(S: String): Boolean;

var I: Integer;

begin

 for I:=1 to NVN do

  if S=VN[I] then begin

   Result:=true;

   exit;

  end;

 Result:=false;

end;

procedure TForm1.DeleteUnavailables;

var I, J, K, N1, N2: Integer; S1, S2: TVN;

function PosVN(V: String; VN: TVN; N: Integer): Integer;

var I: Integer;

begin

 for I:=1 to N do

  if V=VN[I] then begin

   PosVN:=I;

   exit;

  end;

 PosVN:=0;

end;

begin

 N1:=1;

 N2:=1;

 S1[N1]:=S0;

 S2[N2]:=S0;

 while TRUE do begin

  for I:=1 to N1 do begin

   J:=PosVN(S1[I],VN,NVN);

   for K:=1 to Length(Edit2.Text) do

    if (P[J,K]<>'') And (PosVN(P[J,K],S2,N2)=0) then begin

     Inc(N2);

     S2[N2]:=P[J,K];

    end;

  end;

  if N1=N2 then break;

  N1:=N2;

  for I:=1 to N2 do S1[I]:=S2[I];

 end;

 I:=1;

 while (I<=NVN) do

  if PosVN(VN[I],S1,N1)=0 then begin

   Dec(NVN);

   for J:=I to NVN do begin

    Final[J]:=Final[J+1];

    VN[J]:=VN[J+1];

    for K:=1 to Length(Edit2.Text) do P[J,K]:=P[J+1,K];

   end;

  end

  else

   Inc(I);

end;

procedure TForm1.DKA;

var I, J, K, L: Integer; FDo: Boolean;

begin

 FDo:=true;

 while (FDo) do begin

  FDo:=false;

  I:=1;

  while ((Not FDo) And (I<=NVN)) do begin

   J:=1;

   while ((Not FDo) And (J<=Length(Edit2.Text))) do

    if (P[I,J]<>'') And (Not CheckState(P[I,J])) then FDo:=true

                                                 else Inc(J);

   if Not FDo then Inc(I);

  end;

  if FDo then begin

   Inc(NVN);

   VN[NVN]:=P[I,J];

   for I:=1 to Length(VN[NVN]) do begin

    L:=Pos(VN[NVN][I],AVN);

    Final[NVN]:=Final[NVN] Or Final[L];

    for K:=1 to Length(Edit2.Text) do

     if P[L,K]<>'' then P[NVN,K]:=AddVN(P[NVN,K],P[L,K]);

   end;

  end;

 end;

end;

procedure TForm1.Out;

var I, J: Integer;

begin

 Edit4.Text:='';

 Edit6.Text:=S0;

 Edit7.Text:='';

 for I:=0 to StringGrid1.RowCount-1 do StringGrid1.Rows[I].Clear;

 for J:=1 to Length(Edit2.Text) do StringGrid1.Cells[J,0]:=Edit2.Text[J];

 for I:=1 to NVN do begin

  StringGrid1.Cells[0,I]:=VN[I];

  for J:=1 to Length(Edit2.Text) do StringGrid1.Cells[J,I]:=P[I,J];

  if Final[I] then Edit7.Text:=Edit7.Text+VN[I];

 end;

 for I:=1 to NVN do Edit4.Text:=Edit4.Text+VN[I];

end;

function TForm1.AddAVN(var V: String): Boolean;

var I: Integer;

begin

 V:='A';

 for I:=1 to 25 do

  if Pos(V[1],AVN)=0 then begin

   AVN:=AVN+V;

   Result:=true;

   exit;

  end

  else

   Inc(V[1]);

 ShowMessage('Ошибка! Не достаточно нетерминальных символов для построения автоматной грамматики');

 Result:=false;

end;

function TForm1.Auto: Boolean;

var I, J, N: Integer; S, V, VNew: String;

begin

 for I:=1 to 26 do begin

  Final[I]:=false;

  VN[I]:='';

  for J:=1 to 10 do P[I,J]:='';

 end;

 NVN:=Length(Edit1.Text);

 S0:=Edit3.Text;

 if Edit9.Text='' then begin

  ShowMessage('Ошибка! Не задан пустой символ');

  Result:=false;

  exit;

 end;

 if Pos(Edit9.Text,Edit1.Text)>0 then begin

  ShowMessage('Ошибка! Пустой символ найден в нетерминальных символах');

  Result:=false;

  exit;

 end;

 if Pos(Edit9.Text,Edit2.Text)>0 then begin

  ShowMessage('Ошибка! Пустой символ найден в терминальных символах');

  Result:=false;

  exit;

 end;

 for I:=1 to Length(Edit1.Text) do

  if Not (Edit1.Text[I] in ['A'..'Z']) then begin

   ShowMessage('Ошибка! В нетерминальных символах допустимы только заглавные латинские символы');

   Result:=false;

   exit;

  end;

 for I:=1 to Length(Edit2.Text) do

  if Edit2.Text[I] in ['A'..'Z'] then begin

   ShowMessage('Ошибка! В нетерминальных символах не допустимы заглавные латинские символы');

   Result:=false;

   exit;

  end;

 if Pos(S0,Edit1.Text)=0 then begin

  ShowMessage('Ошибка! Целоевого символа грамматики нет в нетерминальных символах');

  Result:=false;

  exit;

 end;

 for I:=1 to Length(Edit1.Text)-1 do

  if Pos(Edit1.Text[I],Copy(Edit1.Text,I+1,Length(Edit1.Text)))<>0 then begin

   ShowMessage('Ошибка! Повтор нетерминального символа');

   Result:=false;

   exit;

  end;

 for I:=1 to Length(Edit2.Text)-1 do

  if Pos(Edit2.Text[I],Copy(Edit2.Text,I+1,Length(Edit2.Text)))<>0 then begin

   ShowMessage('Ошибка! Повтор терминального символа');

   Result:=false;

   exit;

  end;

 for I:=1 to NVN do VN[I]:=Edit1.Text[I];

 AVN:=Edit1.Text;

 ListBox1.Items.Clear;

 for I:=0 to Memo1.Lines.Count-1 do begin

  S:=Memo1.Lines[I];

  if S='' then continue;

  N:=Pos(S[1],Edit1.Text);

  if N=0 then begin

   ShowMessage('Ошибка! Нетерминальный символ не определен: '+Memo1.Lines[I]);

   Result:=false;

   exit;

  end;

  if Pos('=',S)<>2 then begin

   ShowMessage('Ошибка! Отсутствует =: '+Memo1.Lines[I]);

   Result:=false;

   exit;

  end;

  Delete(S,1,2);

  if (Length(S)=1) And (Pos(S[1],Edit1.Text)>0) then begin

   ShowMessage('Ошибка! Недопустимое правило: '+Memo1.Lines[I]);

   Result:=false;

   exit;

  end;

  for J:=1 to Length(Edit1.Text) do

   if Pos(Edit1.Text[J],Copy(S,2,Length(S)-1))>0 then begin

    ShowMessage('Ошибка! Недопустимое правило в ЛЛ-грамматике: '+Memo1.Lines[I]);

    Result:=false;

    exit;

   end;

  for J:=1 to Length(S) do

   if (S[J]=Edit9.Text) And (Length(S)>1) then begin

    ShowMessage('Ошибка! Пустое правило должно состоять из одного символа: '+Memo1.Lines[I]);

    Result:=false;

    exit;

   end

   else

    if (S[J]<>Edit9.Text) And (Pos(S[J],Edit1.Text)=0) And (Pos(S[J],Edit2.Text)=0) then begin

     ShowMessage('Ошибка! Посторонний символ в правиле: '+Memo1.Lines[I]);

     Result:=false;

     exit;

    end;

  V:=Memo1.Lines[I][1];

  while Length(S)>2 do begin

   if Not AddAVN(Vnew) then begin

    Result:=false;

    exit;

   end;

   ListBox1.Items.Add(V+'='+VNew+S[Length(S)]);

   Delete(S,Length(S),1);

   V:=VNew;

  end;

  if (Length(S)>1) And (Pos(S[1],Edit2.Text)>0) then begin

   if Not AddAVN(Vnew) then begin

    Result:=false;

    exit;

   end;

   ListBox1.Items.Add(V+'='+VNew+S[2]);

   ListBox1.Items.Add(VNew+'='+S[1]);

  end

  else

   ListBox1.Items.Add(V+'='+S);

 end;

 Result:=true;

end;

function TForm1.NKA: Boolean;

var I, J, K, N: Integer; S, St: String;

begin

 Final[Pos(Edit3.Text,AVN)]:=true;

 if Not AddAVN(S0) then begin

  Result:=false;

  exit;

 end;

 St:=S0;

 for I:=0 to ListBox1.Items.Count-1 do begin

  S:=ListBox1.Items[I];

  J:=Pos(S[Length(S)],Edit2.Text);

  if Length(S)=3 then

   if S[3]=Edit9.Text then

    if S[1]=Edit3.Text then Final[Pos(S0,AVN)]:=true

                       else St:=St+S[1]

   else begin

    N:=Pos(S0,AVN);

    P[N,J]:=P[N,J]+S[1];

   end

  else begin

   N:=Pos(S[3],AVN);

   P[N,J]:=P[N,J]+S[1];

  end;

 end;

 if St<>S0 then begin

  if Not AddAVN(S0) then begin

   Result:=false;

   exit;

  end;

  N:=Length(AVN);

  for I:=1 to Length(St) do begin

   K:=Pos(St[I],AVN);

   for J:=1 to Length(Edit2.Text) do P[N,J]:=P[N,J]+P[K,J];

  end;

 end;

 NVN:=Length(AVN);

 for I:=1 to NVN do VN[I]:=AVN[I];

 Result:=true;

end;

function TForm1.SortString(S: String): String;

var I, J: Integer; Ch: Char;

begin

 for I:=1 to Length(S)-1 do

  for J:=I+1 to Length(S) do

   if (((S[I]>S[J]) And (S[I]<>S0[1])) Or (S[J]=S0[1])) then begin

    Ch:=S[J];

    S[J]:=S[I];

    S[I]:=Ch;

   end;

 SortString:=S;

end;

procedure TForm1.N4Click(Sender: TObject);

begin

 Button1.Enabled:=false;

 if (Not Auto) Or (Not NKA) then exit;

 Out;

end;

procedure TForm1.N2Click(Sender: TObject);

begin
 ShowMessage('Тема: Построение ДКА, эквивалентного леволинейной грамматике');

end;

procedure TForm1.N3Click(Sender: TObject);

begin

 ShowMessage('Автор: Пугаева Ксения Евгеньевна');

end;

procedure TForm1.N5Click(Sender: TObject);

var NT: String; I, J, K, L: Integer;

begin

 if (Not Auto) Or (Not NKA) then exit;

 Out;

 if Not CheckDKA then begin

  DKA;

  DeleteUnavailables;

  Out;

 end;

 NT:='A';

 for I:=1 to NVN do

  if Length(VN[I])>1 then begin

   J:=1;

   while J<=NVN do

    if NT=VN[J] then begin

     J:=1;

     NT:=Chr(Ord(NT[1])+1);

    end

    else

     Inc(J);

   for K:=1 to 26 do

    for L:=1 to 10 do

     if P[K,L]=VN[I] then P[K,L]:=NT;

   VN[I]:=NT;

   NT:=Chr(Ord(NT[1])+1);

  end;

 Out;

 Button1.Enabled:=true;

end;

procedure TForm1.N6Click(Sender: TObject);

var I, J: Integer; S: String;

begin

 SaveDialog1.InitialDir:=GetCurrentDir;

 if Not SaveDialog1.Execute then exit;

 ListBox1.Items.Clear;

 ListBox1.Items.Add('Q='+Edit4.Text);

 ListBox1.Items.Add('S='+Edit6.Text);

 ListBox1.Items.Add('Q='+Edit7.Text);

 for I:=0 to NVN do with StringGrid1 do begin

  S:='';

  for J:=0 to Length(Edit2.Text) do S:=S+Cells[J,I]+Chr(9);

  ListBox1.Items.Add(S);

 end;

 ListBox1.Items.SaveToFile(SaveDialog1.FileName);

end;

procedure TForm1.N7Click(Sender: TObject);

var I: Integer;

begin

 OpenDialog1.InitialDir:=GetCurrentDir;

 if Not OpenDialog1.Execute then exit;

 ListBox1.Items.LoadFromFile(OpenDialog1.FileName);

 if ListBox1.Items.Count<4 then begin

  N9Click(Sender);

  exit;

 end;

 Memo1.Lines.Clear;

 Edit1.Text:=ListBox1.Items[0];

 Edit2.Text:=ListBox1.Items[1];

 Edit3.Text:=ListBox1.Items[2];

 for I:=3 to ListBox1.Items.Count-1 do

  Memo1.Lines.Add(ListBox1.Items.Strings[I]);

end;

procedure TForm1.N9Click(Sender: TObject);

begin

 ShowMessage(

  '           Входной файл ЛЛ-грамматики'+Chr(13)+Chr(13)+

  'Множество нетерминальных символов'+Chr(13)+

  'Символы алфавита языка'+Chr(13)+

  'Целевой символ грамматики'+Chr(13)+

  'Правила грамматики: нетерминал=правая часть правила'+Chr(13)+Chr(13)+

  'Пример описания ЛЛ-грамматике в файле "ll.txt"'

 );

end;

procedure TForm1.Button1Click(Sender: TObject);

var I, J: Integer; S: String;

begin

 ListBox2.Items.Clear;

 S:=Edit5.Text;

 I:=Pos(Edit6.Text,Edit4.Text);

 while (S<>'') do begin

  ListBox2.Items.Add('('+Edit4.Text[I]+','+S+')->');

  J:=Pos(S[1],Edit2.Text);

  if J=0 then begin

   ListBox2.Items.Add('В цепочке посторонний символ');

   exit;

  end;

  if P[I,J]='' then begin

   ListBox2.Items.Add('Нет перехода из '+Edit4.Text[I]+' по символу '+S[1]);

   exit;

  end;

  I:=Pos(P[I,J],Edit4.Text);

  Delete(S,1,1);

 end;

 ListBox2.Items.Add('('+Edit4.Text[I]+','+Edit9.Text+')');

 if Pos(Edit4.Text[I],Edit7.Text)=0 then

  ListBox2.Items.Add('Цепочка отвергнута, автомат не в конечном состоянии')

 else
  ListBox2.Items.Add('Цепочка прочитана');

end;

end.

5. РЕЗУЛЬТАТЫ РАБОТЫ ПРОГРАММЫ

Результат построения ДКА для примера, рассмотренного в разделе 2, и проверка чтения им цепочки.

Вывод на экран промежуточных результатов в виде НКА, эквивалентного исходной ЛЛ-грамматике.

Пример реакции программы на обнаружение нарушения правил ЛЛ-грамматики.

